Maple-assisted proof of formula for A196742

Robert Israel

6 Oct 2019

There are $5^8 = 390625$ configurations for a 2×4 sub-array, but not all can arise.

We encode these configurations as lists in the order $\begin{bmatrix} x_1 & x_2 & x_3 & x_4 \\ x_5 & x_6 & x_7 & x_8 \end{bmatrix}$ Let $b_0 = 2$, $b_1 = 3$, $b_2 = 1$, $b_3 = 0$, $b_4 = 4$. If $x_i = j$, then the neighbours of site i in the sub-array that have value b_j must be either j or $j - 1$.

```
> b[0]:= 2: b[1]:= 3: b[2]:= 1: b[3]:= 0: b[4]:= 4:
  goodconfig:= proc(x) local t;
     t:= numboccur(b[x[1]], [x[2],x[5]]);
     if t < x[1]-1 or t > x[1] then return false fi;
     t:= numboccur(b[x[2]], [x[1],x[3],x[6]]);
     if t < x[2]-1 or t > x[2] then return false fi;
     t:= numboccur(b[x[3]], [x[2],x[4],x[7]]);
     if t < x[3]-1 or t > x[3] then return false fi;
     t:= numboccur(b[x[4]], [x[3],x[8]]);
     if t < x[4]-1 or t > x[4] then return false fi;
     t:= numboccur(b[x[5]], [x[1],x[6]]);
     if t < x[5]-1 or t > x[5] then return false fi;
     t:= numboccur(b[x[6]], [x[2],x[5],x[7]]);
     if t < x[6]-1 or t > x[6] then return false fi;
     t:= numboccur(b[x[7]], [x[3],x[6],x[8]]);
     if t < x[7]-1 or t > x[7] then return false fi;
     t:= numboccur(b[x[8]], [x[4],x[7]]);
     if t < x[8]-1 or t > x[8] then return false fi;
     true
  end proc:
  Configs:= select(goodconfig, [seq(convert(5^8+i,base,5)[1..8],i=
  0..5^8-1)]):
  nops(Configs);
```

$$1512 \quad \textbf{(1)}$$

There are 1512 allowed configurations.

Consider the 1512×1512 transition matrix T with entries $T_{ij} = 1$ if the first two rows of a 3×4 sub-array could be in configuration i while the last two rows are in configuration j, and 0 otherwise. The following code computes it.

```
> Compatible:= proc(i,j) local Xi,Xj,k;
   Xi:= Configs[i]; Xj:= Configs[j];
   if Xi[5..8] <> Xj[1..4] then return 0 fi;
   if numboccur(b[Xi[5]],[Xi[1],Xi[6],Xj[5]]) <> Xi[5] then return
  0 fi;
   if numboccur(b[Xi[6]],[Xi[2],Xi[5],Xi[7],Xj[6]]) <> Xi[6] then
  return 0 fi;
   if numboccur(b[Xi[7]],[Xi[3],Xi[6],Xi[8],Xj[7]]) <> Xi[7] then
```

```
return 0 fi;
 if numboccur(b[Xi[8]],[Xi[4],Xi[7],Xj[8]]) <> Xi[8] then 0 else
1 fi;
end proc:
T:= Matrix(1512,1512,Compatible):
```

Thus for $n \ge 2$ $a(n) = \frac{u^T T^{n-2} v}{2}$ where u is a column vector with 1 for configurations whose first row could be a top row, 0 otherwise, and similarly v has 1 for configurations whose second row could be a bottom row.

```
> u:= Vector(1512,proc(i) local x; x:= Configs[i];
     if numboccur(b[x[1]], [x[2],x[5]])=x[1]
     and numboccur(b[x[2]],[x[1],x[3],x[6]]) = x[2]
     and numboccur(b[x[3]],[x[2],x[4],x[7]]) = x[3]
     and numboccur(b[x[4]],[x[3],x[8]]) = x[4] then 1 else 0 fi
  end proc):
  v:= Vector(1512,proc(i) local x; x:= Configs[i];
     if numboccur(b[x[5]], [x[1],x[6]]) = x[5]
     and numboccur(b[x[6]],[x[2],x[5],x[7]]) = x[6]
     and numboccur(b[x[7]],[x[3],x[6],x[8]]) = x[7]
     and numboccur(b[x[8]],[x[4],x[7]]) = x[8] then 1 else 0 fi
   end proc) :
```

To check, here are the first few entries of our sequence (apart from a_1, which doesn't really fit the pattern, although it does work with the recurrence).

```
> Tv[0]:= v:
  for n from 1 to 32 do Tv[n]:= T . Tv[n-1] od:
> A:= [seq(u^%T . Tv[n],n=0..32)];
```

A := [7, 71, 401, 2923, 19567, 134507, 919377, 6283897, 42987489, 293903925, 2009922151, 13744157339, 93985887721, 642701573055, 4394944856905, 30053778082923, 205515234410701, 1405365233289259, 9610242269541951, 65717262620421205, 449391242181726185, 3073050779567818717, 21014297307438194435, 143701071568534457665, 982664215871735137459, 6719706057700845919781, 45951046927195676789363, 314224862776603781877735, 2148748961894868779941811, 14693688018580354100210337, 100479149223628856949536391, 687101796087103520553493563, 4698575593385785360547261561] **(2)**

Now here is the empirical recurrence formula. It says that $u^T T^n Q(T)\, v = 0$ for all nonnegative integers n, where Q is the following polynomial.

```
> n:= 'n': empirical:= a(n) = 3*a(n-1) +26*a(n-2) +16*a(n-3) -99*a
  (n-4) +2*a(n-5) +81*a(n-6) -330*a(n-7) +340*a(n-8) -212*a(n-9)
  +124*a(n-10) +81*a(n-11) -30*a(n-12) +5*a(n-13) -3*a(n-14) -a
  (n-15);
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(15-i),i=0.
  .15),t);
```

$$empirical := a(n) = 3\,a(n-1) + 26\,a(n-2) + 16\,a(n-3) - 99\,a(n-4) + 2\,a(n-5) + 81\,a(n-6) - 330\,a(n-7) + 340\,a(n-8) - 212\,a(n-9) + 124\,a(n-10) + 81\,a(n-11) - 30\,a(n-12) + 5\,a(n-13) - 3\,a(n-14) - a(n-15)$$

$$Q := t \mapsto t^{15} - 3\,t^{14} - 26\,t^{13} - 16\,t^{12} + 99\,t^{11} - 2\,t^{10} - 81\,t^{9} + 330\,t^{8} - 340\,t^{7} + 212\,t^{6} - 124\,t^{5} - 81\,t^{4} + 30\,t^{3} - 5\,t^{2} + 3\,t + 1 \quad \textbf{(3)}$$

To complete the proof, we verify that $Q(T)\ v = 0$.

```
> Qv:= add(coeff(Q(t),t,j)*Tv[j],j=0..15):
> Qv^%T . Qv;
```

$$0 \tag{4}$$